
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc114650956]Corrections required
1.1 [bookmark: _Toc114650957]Correction to the function fl_ReleaseCallWithOptionalDeregisterOrRelease
	Template name
	fl_ReleaseCallWithOptionalDeregisterOrRelease

	Reason for change
	While running testcase 11.4.9, fl_TC_11_4_9_NR5GC_TestBody fl_ReleaseCallWithOptionalDeregisterOrRelease f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); is called, and after this the NR5GC PTC waits for a coordination message from IMS PTC.

On the IMS PTC side, f_TC_11_4_9_NR5GC_IMS2 f_IMS_EmergencyCallAndReleaseWithoutRegistration f_IMS_EmergencyCallRelease is called once the coordination message from NR5GC PTC to release call is received.
This function then calls f_IMS_EmergencyCallRelease f_IMS_CallReleaseMO a_IMS_ReceiveByeSendOK.

Thus, until the UE sends a BYE, and the SS has sent a 200OK followed by TCP close, the call f_TC_11_4_9_NR5GC_IMS2 f_IMS_EmergencyCallAndReleaseWithoutRegistration f_IMS_IPCAN_SendCoOrdMsg(IPCAN, cms_IMS_IPCAN_Trigger("IMS Call Release Msgs Sent")); will not happen.

The problem here is that as soon as the UE sends a BYE, it will also send a PDU Session Release REQ which needs to be handled in the NR5GC PTC. But at this stage, the NR5GC PTC is waiting for a coordination message "IMS Call Release Msgs Sent".
This results in the PDU Session Release REQ not being handled which causes testacse failure.

This needs to be addressed.

	Summary of change
	Moved the reception of coordination message IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent")); to the same level as reception of PDU Session Release REQ in function fl_ReleaseCallWithOptionalDeregisterOrRelease.

Please see below for detailed change.

	TTCN module
	Emergency_NR5GC.ttcn

	MCC160 Comment
	Accepted



    
Before Change:
	function fl_ReleaseCallWithOptionalDeregisterOrRelease (NR_CellId_Type p_NR_CellId)  runs on NR5GC_PTC {
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var O1_Type v_PDUSessionId := int2oct(f_Get_PDUSessionIdForDNNType(v_PDUSessionInfoList, Emergency_PDN), 1);
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_RxdMsg;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var boolean v_WaitForReleaseComplete := false;
    timer t_Wait := 5.0;   // @sic R5-223439 sic@

    //@siclog "Step 5" siclog@
    //Make the UE release the emergency call.
    // @sic R5-223439 sic@
    //@siclog "Step 6A-6C" siclog@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); // @sic R5-225362 sic@
    IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"));  // @sic R5-225362 sic@ sic@

    //@siclog "Step 7" siclog@
    t_Wait.start; // @sic R5s220560 sic@
    //EXCEPTION: In parallel to step 8 below the steps specified in the parallel behaviours may take place
    alt {
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
         v_RxdMsg := v_ReceivedAsp.Signalling.Nas[0].Pdu;
         // The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
         if (ischosen(v_RxdMsg.Msg.deregistration_RequestMO)) {
           if (not f_Check_NG_DeregistrationReqMOMsg (v_RxdMsg.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request failed");
           }
           //The SS replies with a DEREGISTRATION ACCEPT message.
           SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
         }

         // @sic R5-223439 sic@
         // The UE transmits a PDU SESSION RELEASE REQUEST message
         else if (ischosen(v_RxdMsg.PiggybackedPduList[0].Msg.pdu_Session_Release_Request)) {
           if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfoList, v_RxdMsg, v_PDUSessionId)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
          }
           // set flag to expect complete
           v_WaitForReleaseComplete := true;
           // Complete the release procedure
           v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionId, cs_GMM_GSM_Cause (omit, '00011010'B));
           SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
         }
       
         // The UE transmits a PDU SESSION RELEASE COMPLETE message?
         else if (v_WaitForReleaseComplete and ischosen(v_RxdMsg.PiggybackedPduList[0].Msg.pdu_Session_Release_Complete)) {
          if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_RxdMsg, v_PDUSessionId)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
          }
          v_WaitForReleaseComplete := false;
         }
         else {
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Wrong NAS Message Received");
         }
         
         // Keep going round until timer expires @sic R5-223439 sic@
         repeat;
       }
    
      //@siclog "Step 8" siclog@
      [] t_Wait.timeout  //@sic R5-216198 sic@
       { // if Release Complete hasn't arrived before timeout, then need to wait for it so it doesn't cause an error when it does arrive
         if (v_WaitForReleaseComplete) {
           SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
           if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId)) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
           }
         }
       }
    }
 }



After Change:
	function fl_ReleaseCallWithOptionalDeregisterOrRelease (NR_CellId_Type p_NR_CellId)  runs on NR5GC_PTC {
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var O1_Type v_PDUSessionId := int2oct(f_Get_PDUSessionIdForDNNType(v_PDUSessionInfoList, Emergency_PDN), 1);
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_RxdMsg;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var boolean v_WaitForReleaseComplete := false;
    timer t_Wait := 5.0;   // @sic R5-223439 sic@

    //@siclog "Step 5" siclog@
    //Make the UE release the emergency call.
    // @sic R5-223439 sic@
    //@siclog "Step 6A-6C" siclog@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2], cms_IMS_IPCAN_Trigger("Trigger MO Call Release")); // @sic R5-225362 sic@
    //WA#11_4_9 MOVED IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"));  // @sic R5-225362 sic@ sic@

    //@siclog "Step 7" siclog@
    t_Wait.start; // @sic R5s220560 sic@
    //EXCEPTION: In parallel to step 8 below the steps specified in the parallel behaviours may take place
    alt {
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
       {
         v_RxdMsg := v_ReceivedAsp.Signalling.Nas[0].Pdu;
         // The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
         if (ischosen(v_RxdMsg.Msg.deregistration_RequestMO)) {
           if (not f_Check_NG_DeregistrationReqMOMsg (v_RxdMsg.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request failed");
           }
           //The SS replies with a DEREGISTRATION ACCEPT message.
           SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
         }

         // @sic R5-223439 sic@
         // The UE transmits a PDU SESSION RELEASE REQUEST message
         else if (ischosen(v_RxdMsg.PiggybackedPduList[0].Msg.pdu_Session_Release_Request)) {
           if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, v_PDUSessionInfoList, v_RxdMsg, v_PDUSessionId)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
          }
           // set flag to expect complete
           v_WaitForReleaseComplete := true;
           // Complete the release procedure
           v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionId, cs_GMM_GSM_Cause (omit, '00011010'B));
           SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
         }
       
         // The UE transmits a PDU SESSION RELEASE COMPLETE message?
         else if (v_WaitForReleaseComplete and ischosen(v_RxdMsg.PiggybackedPduList[0].Msg.pdu_Session_Release_Complete)) {
          if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_RxdMsg, v_PDUSessionId)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
          }
          v_WaitForReleaseComplete := false;
         }
         else {
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Wrong NAS Message Received");
         }
         
         // Keep going round until timer expires @sic R5-223439 sic@
         repeat;
       }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger ("IMS Call Release Msgs Sent"))  // @sic R5-225362 sic@ sic@ //WA#11_4_9
        {// Keep going round until timer expires @sic R5-223439 sic@
         repeat;
        }
    
      //@siclog "Step 8" siclog@
      [] t_Wait.timeout  //@sic R5-216198 sic@
       { // if Release Complete hasn't arrived before timeout, then need to wait for it so it doesn't cause an error when it does arrive
         if (v_WaitForReleaseComplete) {
           SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
           if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfoList, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionId)) {
             f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Complete failed");
           }
         }
       }
    }
 }



